
Вступление

Типичная схемная реализация автомата содержит:
I параллельный регистр R, хранящий состояние автомата
I комбинационную схему ΣB , реализующую функцию выхода
I комбинационную схему ΣT , реализующую функцию переходов

Обсудим то, как выглядит такая типичная реализация на языке Verilog

Автомат → V: состояния

Значение на выходе R в каждый момент времени — текущее состояние

Значение на входе R в каждый момент времени — следующее состояние:
по переднему фронту тактового сигнала это состояние
сохраняется в регистре и становится текущим

Объявим соответствующие точки state и next_state

Например, для A наименьшая ширина этих точек — dlog23e = 2:

reg [1:0] state, next_state;

(next_state может быть и соединением —
смотря как реализована схема ΣT )

Автомат → V: состояния

Чтобы избежать опечаток, связанных с нумерацией состояний,
присвоим каждому состоянию автомата уникальное «наглядное» имя
Это имя не имеет смысла для пользователя и должно быть недоступно
для изменения им, и для задания таких имён хорошо подходят
локальные параметры:

localparam S_TOP = 0,
S_RIGHT = 1,
S_LEFT = 2;

Реализуем регистр R типовым способом:
always @(posedge clk, posedge rst)
if(rst) state <= S_TOP;
else state <= next_state;

Автомат → V: переходы

Реализуем функцию переходов (ΣT ) в отдельной процедуре:

always @* begin
next_state = 1’bx; // Все случаи, не перечисленные дальше, неважны
case(state) // Для каждого состояния перечи́слим исходящие из него ду́ги

// Ду́ги символьного автомата над подходящими предикатами
// легко переписываются как код

S_TOP: if(x) next_state = S_RIGHT;
else next_state = S_TOP;

S_RIGHT: next_state = S_LEFT;
S_LEFT: if(x) next_state = S_TOP;

else next_state = S_LEFT;
endcase

end

Автомат → V: выход

Реализуем функцию выхода (ΣB) в отдельной процедуре:
always @* begin

y1 = 1’bx; y2 = 1’bx; // Все случаи, не перечисленные дальше, неважны
case(state) // Реализуем функцию выхода таблично
S_TOP, S_RIGHT: begin
y1 = 0;
y2 = 1;

end
S_LEFT: begin
y1 = 1;
y2 = 0;

end
endcase

end

Автомат → V: всё вместе

reg [1:0] state, next_state;
localparam S_TOP = 0, S_RIGHT = 1, S_LEFT = 2;
always @(posedge clk, posedge rst)

if(rst) state <= S_TOP;
else state <= next_state;

always @* begin
next_state = 1’bx;
case(state)
S_TOP: if(x) next_state = S_RIGHT; else next_state = S_TOP;
S_RIGHT: next_state = S_LEFT;
S_LEFT: if(x) next_state = S_TOP; else next_state = S_LEFT;
endcase

end
always @* begin

y1 = 1’bx; y2 = 1’bx;
case(state)
S_TOP, S_RIGHT: begin y1 = 0; y2 = 1; end
S_LEFT: begin y1 = 1; y2 = 0; end
endcase

end

Автомат → V: другие варианты (переходы)

Не всегда бывает удобно начинать процедуру ΣT
с присваиваний «по умолчанию значение неважно»

Например, если в автомате много петель:

always @* begin
next_state = state; // По умолчанию не изменяем состояние
case(state) // Перечисляем все случаи изменения состояния
S_TOP: if(x) next_state = S_RIGHT;
S_RIGHT: next_state = S_LEFT;
S_LEFT: if(x) next_state = S_TOP;
endcase

end

Автомат → V: другие варианты (переходы)

A:

y1/0, y2/1

y1/0, y2/1y1/1, y2/0

¬x

x

1¬x

x ΣT R ΣBxn
. . .x1

ym
. . .

y1

statenext_state

Другой пример — если в автомате много переходов в заданное
состояние:

always @* begin
next_state = S_TOP; // По умолчанию переходим в выделенное состояние
case(state) // Перечисляем все случаи переходов в другие состояния
S_TOP: if(x) next_state = S_RIGHT;
S_RIGHT: next_state = S_LEFT;
S_LEFT: if(!x) next_state = S_LEFT;
endcase

end

Автомат → V: другие варианты (выходы)

Не всегда бывает удобно начинать процедуру ΣB
с присваиваний «по умолчанию выходы произвольны»

Например, если в автомате есть «преобладающие» выходы:

always @* begin
y1 = 0; y2 = 1; // Преобладающие выходные значения
if(state == S_LEFT) begin // Редкие выходные значения
y1 = 1;
y2 = 0;

end
end

Автомат → V: другие варианты (выходы)

Другой пример: если разнообразие выходных значений невелико,
то комбинационные выражения могут оказаться нагляднее процедур:

assign y1 = (state == S_LEFT);
assign y2 = !y1;

Автомат → V: другие варианты (всё вместе)

A:

y1/0, y2/1

y1/0, y2/1y1/1, y2/0

¬x

x

1¬x

x ΣT R ΣBxn
. . .x1

ym
. . .

y1

statenext_state

Если кажется, что схемы ΣT , ΣB в двух разных процедурах ненаглядны,
то ничто не запрещает совместить эти процедуры (не переусердствуя):

always @* begin
next_state = state; y1 = 0; y2 = 1;
case(state)
S_TOP: if(x) next_state = S_RIGHT;
S_RIGHT: next_state = S_LEFT;
S_LEFT: begin
if(x) next_state = S_TOP;
y1 = 1;
y2 = 0;

end
endcase

end

(«Наглядность» — субъективное понятие,
лишь бы только было проще избежать глупых ошибок)

